Linear Algebra in Maple

You **must** load the linear algebra package.

```
> with(linalg):
```

```
Warning, new definition for norm

Warning, new definition for trace
```

```
> A := matrix([[1,I],[-I,1]]);
```

$$A := \begin{bmatrix} 1 & I \\ -I & 1 \end{bmatrix}$$

Use the command **conjugate** to conjugate the entries of the matrix. You have to use **evalm** (evaluation in matrices) for operations on matrices to be executed.

```
> evalm(conjugate(A));
```

$$\begin{bmatrix} 1 & -I \\ I & 1 \end{bmatrix}$$

The command **htranspose** calculates the conjugate transpose (adjoint) of a matrix.

```
> htranspose(A);
```

$$\begin{bmatrix} 1 & I \\ -I & 1 \end{bmatrix}$$

The comand **map** applies a procedure to each operand of an expression. Use it, for example, to differentiate entries of a matrix.

Here is the matrix of problem 27, p. 353.

```
> psi :=
> matrix([[exp(t),exp(-2*t),exp(3*t)],[-4*exp(t),-exp(-2*t),2*exp(3*t)],
> [-exp(t),-exp(-2*t),exp(3*t)]]);
```

$$\psi := \begin{bmatrix} e^t & e^{(-2\,t)} & e^{(3\,t)} \\ -4\,e^t & -e^{(-2\,t)} & 2\,e^{(3\,t)} \\ -e^t & -e^{(-2\,t)} & e^{(3\,t)} \end{bmatrix}$$

We calculate the derivative of ψ.

```
> map(diff,psi,t);
```

$$\begin{bmatrix} e^t & -2\,e^{(-2\,t)} & 3\,e^{(3\,t)} \\ -4\,e^t & 2\,e^{(-2\,t)} & 6\,e^{(3\,t)} \\ -e^t & 2\,e^{(-2\,t)} & 3\,e^{(3\,t)} \end{bmatrix}$$

We verify that ψ satisfies the differential equation in problem 27.

```
> B := matrix([[1,-1,4],[3,2,-1],[2,1,-1]]);
```

$$B := \begin{bmatrix} 1 & -1 & 4 \\ 3 & 2 & -1 \\ 2 & 1 & -1 \end{bmatrix}$$

Matrix multiplication is indicated by **&*.**

```
> evalm(B&*psi);
```

$$\begin{bmatrix} e^t & -2\,e^{(-2\,t)} & 3\,e^{(3\,t)} \\ -4\,e^t & 2\,e^{(-2\,t)} & 6\,e^{(3\,t)} \\ -e^t & 2\,e^{(-2\,t)} & 3\,e^{(3\,t)} \end{bmatrix}$$

This is the same as the derivative of ψ.

Similarly, we can use map to integrate. For example, to integrate each term of ψ from 0 to 1:

```
> map(int,psi,t=0..1);
```

$$\begin{bmatrix} e-1 & -\frac{1}{2}\,e^{(-2)}+\frac{1}{2} & \frac{1}{3}\,e^3-\frac{1}{3} \\ -4\,e+4 & \frac{1}{2}\,e^{(-2)}-\frac{1}{2} & \frac{2}{3}\,e^3-\frac{2}{3} \\ -e+1 & \frac{1}{2}\,e^{(-2)}-\frac{1}{2} & \frac{1}{3}\,e^3-\frac{1}{3} \end{bmatrix}$$

The command **basis** lets you find a basis for the span of a set of vectors. You can use the command **vector** to create the vectors. For example, here are the vectors of problem 10 on p. 363.

```
> x[1] := vector([1,2,-2]); x[2] := vector([3,1,0]); x[3]:=
> vector([2,-1,1]); x[4] := vector([4,3,-2]);
```

$$x_1 := [1,\ 2,\ -2]$$

$$x_2 := [3,\ 1,\ 0]$$

$$x_3 := [2,\ -1,\ 1]$$

$$x_4 := [4,\ 3,\ -2]$$

```
> basis([x[1],x[2],x[3],x[4]]);
```

$$[x_1,\ x_2,\ x_3]$$

To write x_4 as a linear combination of x_1, x_2, x_3:

```
> v := evalm(x[4]-a*x[1]-b*x[2]-c*x[3]);
```

$$v := [4-a-3\,b-2\,c,\ 3-2\,a-b+c,\ -2+2\,a-c]$$

We need to extract the components of v, then set them equal to 0 and solve the resulting equations. The ith component is v[i].

```
> solve({v[1],v[2],v[3]},{a,b,c});
```

The command **eigenvects** allows you to calculate the eigenvectors of a matrix. The answer comes in the form of a list. Each item in the list consists of an eigenvector, its multiplicity, and a basis for the eigenspace. As an example, here is problem 17 on p. 363.

```
> C := matrix([[-2,1],[1,-2]]);
```

$$C := \begin{bmatrix} -2 & 1 \\ 1 & -2 \end{bmatrix}$$

```
> e := eigenvects(C);
```

$$e := [-1, 1, \{[1, 1]\}], [-3, 1, \{[-1, 1]\}]$$

To make a matrix T whose columns are the eigenvectors we just found, we first have to extract the eigenvectors. For example, [-1,1] is the first thing in {[-1,1]}, which is the third in the second element of e, i.e., it is e[2][3][1].

```
> T := transpose(matrix([e[1][3][1],e[2][3][1]]));
```

$$T := \begin{bmatrix} 1 & -1 \\ 1 & 1 \end{bmatrix}$$

The command **diag** produces a diagonal matrix with the given entries on the diagonal.

```
> Diag := diag(-1,-3);
```

$$Diag := \begin{bmatrix} -1 & 0 \\ 0 & -3 \end{bmatrix}$$

The command **inverse** finds the inverse of a matrix. We use it in checking that $T^{(-1)}\, C\, T = Diag$.

```
> evalm(inverse(T)&*C&*T);
```

$$\begin{bmatrix} -1 & 0 \\ 0 & -3 \end{bmatrix}$$

Warning: You cannot use either Psi or D as a name in Maple, because both are built in.